

CS061 – Lab 5

Building Subroutines

1 High Level Description

Today you will learn the art of building subroutines, and become invincible.

2 Our Objectives for This Week

1. If you have not done so already, complete Lab 4, exercise 4
(Go back to your your lab 4 directory, and follow the instructions in your lab 4 specs)
You should have completed this step BEFORE the start of your lab 5 session)
2. Lab 4 review, & intro to subroutines – Exercise 1
3. Roll your own subs – Exercises 2 - 3
4. Feel smug about it all -- Exercise 4

3. Subroutines: The Art of writing something *once*

Subroutines are a bit like functions in C++, they allow you to "package" your code for re-use, or for repeated use, or simply to make your code more readable.

Here is the basic structure of every subroutine you will ever write in LC3:

```
=====
; Subroutine: SUB_intelligent_name_goes_here_3200
; Parameter: (Register you are "passing"): [description of parameter]
; Postcondition: [a short description of what the subroutine accomplishes]
; Return Value: [which register (if any) has a return value and what it means]
=====

                .orig x3200          ; use the starting address as part of the sub name
=====
; Subroutine Instructions
=====

; (1) Backup R7 and any registers that this subroutine changes except for Return Values
; (2) Whatever algorithm this subroutine is intended to perform - only ONE task per sub!!
; (3) Restore the registers that you backed up
; (4) RET - return to the instruction following the subroutine invocation

=====
; Subroutine Data
=====
BACKUP_R0_3200  .BLKW #1 ; Make one of these for each register that the subroutine changes
BACKUP_R7_3200  .BLKW #1 ; ... EXCEPT for Return Value(s)
```

The header contains information you will need when reusing the subroutine:

- **Subroutine name:**
Give the subroutine a good name and append the subroutine's address to it to make it unique. For example, if the subroutine "SUB_PRINT_ARRAY" starts at x3600, you should name the subroutine "SUB_PRINT_ARRAY_3600" to make it completely unique.
- **Parameters:**
Any parameters that you pass to the subroutine. This is a bit like passing params into a C++ function, except here you pass them in via specific registers rather than named variables.
- **Postcondition:**
What the subroutine actually does so you won't have to guess later ...
- **Return Value:**
The register(s) in which the subroutine returns its result (if any). Unlike C++ functions, you can return multiple values from a subroutine, one per register.
If you are really careful you can even return a value in a register that was used to pass in a parameter (*but not until you really know what you're doing!!*)
- **.ORIG value:**
Each subroutine that you write needs to be placed somewhere specific in memory (just like our "main" code which we always locate at x3000).
A good convention to use is {x3200, x3400, x3600, ...} for the {first, second, third, ...} subroutine.

Once your header is done, you can write your subroutine. This is a 4-step process:

1. Backing up registers:

In this step, use ST to backup R7 and any other registers that this subroutine changes except for registers used for passing in parameters and/or for returning values.

You must always backup R7 because - as we will see below - R7 stores the return address, the address of the instruction following the invocation.

Note that BIOS routines (TRAP instructions) are themselves subroutines, and therefore also use R7 to store their return address. So if you use, say, OUT inside your subroutine, and you haven't backed up your own subroutine's R7, the BIOS routine will overwrite R7, and you will not be able to get back to main!

2. Write your subroutine code:

Write whatever code is necessary to make the subroutine do its thing.

IMPORTANT: all prompts & error messages relating to the subroutine should be part of the subroutine - so, for instance, if a subroutine takes input from the user, the corresponding prompt must be in the subroutine data block, not in "main" where the subroutine is invoked.

3. Restore registers:

In this step, use LD to restore the registers that you backed up in step 1.

Remember to always backup/restore R7.

Remember to never backup/restore register(s) that contain your Return Value(s).

4. Return:

Use the RET instruction to return to where you came from - a bit like return in C++ (*RET is not a separate LC3 instruction - it is just an alias for **JMP R7***)

Reminder about Register Transfer Notation:

- Rn = a register
- (Rn) = the contents of that register
- Mem[some address] = the contents of the specified memory address.
- a <-- b = transfer (i.e. copy) the value b to the location a.
 - R5 <-- (R4) means "copy the contents of Register 4 to Register 5, overwriting any previous contents of Register 5" - e.g. ADD R5, R4, #0
 - Mem[xA400] <-- Mem[(R3)]
means "obtain the value stored in R3 and treat it as a memory address; obtain the value stored at that address; copy that value into memory at the address xA400."

Invoking a subroutine

This is a bit like a function call in C++ (but much simpler!)

JSR and JSRR (*two versions of the same instruction, differing only in their memory addressing modes*)

JSR label works just like BRnzp - i.e. it *unconditionally* transfers control to the instruction at label;

JSRR R6 works just like JMP - i.e. it transfers control to the instruction located at the address stored in R6, known as the *base register* (you can use any register as the BaseRegister, but we standardize on R6).

However, there is one very big difference between JSR/JSRR and BR/JMP:

before transferring control with JSR or JSRR, the very first thing that happens is that the address of the next instruction - i.e. the return point - is stored in **R7**.

This means that at the end of the subroutine, we can get back to where we jumped from with RET

In this class, we will be using the **JSRR** mode almost exclusively, as it allows us to invoke subroutines anywhere in memory; we just have to first get the subroutine address in a BaseRegister via a pointer (*the same techniques we used to set up arrays in lab 4: LD into R6 from local data SUB_PTR*).

Example Code: Yay!!!

Here is an example that calls a really short subroutine that takes the 2's complement of (R1). Note that this is a single file, ending with single .END, containing both main and subroutine, each of which has its own .ORIG

```
1 ;=====
2 ; Main:
3 ;     A test harness for SUB_TWOS_COMPLEMENT_3200 subroutine
4 ;=====
5         .orig x3000
6 ; Instructions:
7         ld r1, const          ; get number to test subroutine with
8
9 ; Call the subroutine (get its address, then Jump to Subroutine):
10        ld r6, sub_twos_comp_ptr
11        jsrr r6
12
13        lea r0, completed_msg  ; tell the user the job is done
14        puts
15
16        halt
17
18 ; Local data (Main):
19 const      .fill #29
20 sub_twos_comp_ptr .fill x3200          ; address of subroutine
21 completed_msg .stringz "The 2's comp. of the value in R1 is now available in R2\n"
22
23 ;=====
24 ; subroutine: : SUB_TWOS_COMPLEMENT_3200
25 ; Input (R1): the value whose two's complement will be calculated
26 ;           This value is not modified by the program.
27 ; Postcondition: the subroutine has calculated the twos complement
28 ;               of the value in R1, and stored it in R2
29 ; Return value (R2): the two's complement of the value in R1
30 ;               i.e. R2 <-- -(R1)
31 ;=====
32        .orig x3200
33 ; subroutine instructions:
34
35 ; (1) backup affected registers:
36        st r3, backup_r3_3200      ; r3 is modified by the subroutine.
37        st r7, backup_r7_3200
38 ; We didn't really need to use r3 as a temporary data store
39 ; - we are just using it to illustrate good register backup practices.
40 ; Also, the subroutine does not do any i/o, so we don't really need to back up R7 either
41 ; - but we ALWAYS back up R7 anyway, just in case!
42
43 ; (2) subroutine algorithm:
44        not r3, r1
45        add r3, r3, #1              ; r3 <-- -(r1)
46        add r2, r3, #0              ; r2 <-- (r3) copy r3 to r2
47
48 ; (3) restore backed up registers
49        ld r3, backup_r3_3200
50        ld r7, backup_r7_3200
51
52 ; (4) Return:
53        ret
54
55 ; Local data for subroutine SUB_TWOS_COMPLEMENT_3200:
56 backup_r3_3200 .blkw #1
57 backup_r7_3200 .blkw #1
58
59 ;=====
60 ;           END SUBROUTINE
61 ;=====
62
63        .end                      ; NOTE: just ONE .end for the whole file
```

Next is a slightly longer program that has three different strings (each stored in its own array) and calls a subroutine that prints out the contents of an array – in other words, we call the same subroutine 3 times. The address of the string to print is passed to the subroutine in R1.

This time, the main and the subroutine are in different files. Each file has its own .ORIG and its own .END. You can load both files into the simpl simulator with the command line:

simpl main.asm library.asm *(note that the main routine must be listed first)*

```
L:\Instruments\Vabs\Vab05\subroutine_example_code\Vab5_example2a_code.asm      Tuesday, February 05, 2013 1:02 PM
;-----
; File: main.asm
; Main:
;   A test harness for the SUB_PRINT_ARRAY_3400 subroutine
;-----
                                .ORIG x3000
; Instructions
                                LEA RO, MSG_01
                                PUTS                                ; print description of what we are doing
                                LD RO, NEWLINE
                                OUT                                  ; print a newline

                                LEA R1, ARRAY_1
                                JSR SUB_PRINT_ARRAY_3400              ; R1 <-- addr(ARRAY_1) (subroutine param)
                                                                ; jump to subroutine SUB_PRINT_ARRAY_3400
                                                                ; - possible ONLY if sub is within +/- 1k

                                LD RO, NEWLINE
                                OUT                                  ; print newline

                                LEA R1, ARRAY_2
                                JSR SUB_PRINT_ARRAY_3400              ; print the contents of ARRAY_2
                                LD RO, NEWLINE
                                OUT                                  ; print a newline

                                LEA R1, ARRAY_3
                                JSR SUB_PRINT_ARRAY_3400              ; print the contents of ARRAY_3
                                LD RO, NEWLINE
                                OUT                                  ; print a newline

                                LEA RO, MSG_02
                                PUTS                                ; print "Done."
                                HALT

; Local Data
NEWLINE                        .FILL #10                            ; newline char
MSG_01                         .STRINGZ "Contents of ARRAY:"
MSG_02                         .STRINGZ "Done."
ARRAY_1                        .STRINGZ "Thank you very much!"
ARRAY_2                        .FILL 'S'
                                .FILL 'a'
                                .FILL 'l'
                                .FILL 'a'
                                .FILL 'm'
                                .FILL 'a'
                                .FILL 't'
                                .FILL ' '
                                .FILL 'p'
                                .FILL 'o'
                                .FILL '!'
                                .FILL #0                            ; marks end of string
ARRAY_3                        .STRINGZ "Danke schon!"

                                .END
```

```

;-----
; File: library.asm
;
; Subroutine: PRINT_ARRAY
; Parameter (R1): The addr of the beginning of an array of characters
;                 that is terminated by the null character, #0.
; Postcondition: The subroutine has printed to console all characters
;                 starting at R1 and continuing until a 0 is reached.
; Return Value: None
;-----

                .ORIG x3400

; Subroutine Instructions
; (1) Backup R7 & any registers the subroutine changes except Return Values
SUB_PRINT_ARRAY_3400    ST R0, BACKUP_R0_3400    ; backup R0
                        ST R1, BACKUP_R1_3400    ; backup R1
                        ST R7, BACKUP_R7_3400    ; backup R7 (why?)

; (2) Subroutine's Algorithm
WHILE_LOOP_3400        LDR R0, R1, #0            ; R0 <-- Mem[ (R1) + 0 ]
                        BRz END_WHILE_LOOP_3400 ; if ( (R0) == 0 ) { break out of loop }
                        else...
                        OUT                        ; print (R0)
                        ADD R1, R1, #1            ; R1 <-- (R1) + 1
                        BR WHILE_LOOP_3400        ; goto WHILE_LOOP_3400 no matter what
END_WHILE_LOOP_3400

; (3) Restore registers
                        LD R0, BACKUP_R0_3400    ; restore R0
                        LD R1, BACKUP_R1_3400    ; restore R1
                        LD R7, BACKUP_R7_3400    ; restore R7

; (4) Return
                        RET                        ; PC <-- (R7) i.e. jump to the address (R7)

; Subroutine Data
BACKUP_R0_3400        .BLKW #1
BACKUP_R1_3400        .BLKW #1
BACKUP_R7_3400        .BLKW #1

                .END

```

Note that if you use any other LC-3 emulator than simpl, you will probably have to use the technique of this second example, with a separate .asm file for each routine/subroutine, since the emulator may not be able to handle multiple .orig pseudo-ops in a single file as simpl does.

3.2 Exercises

NOTE: No "ghost writing"! That gets an instant 0!

Exercise 1

Recall that exercise 4 from last week created an array of the first 10 powers of 2 $\{2^0, 2^1, 2^2, \dots, 2^9\}$, and then attempted to print these values to console as though they were ascii characters - with interesting results!

But in programming assignment 3, you figured out how to take the 16-bit value in a register and print to console the corresponding 16 ascii 1's and 0's.

Your first lab 5 exercise is to take your assn 3 code and turn it into a subroutine:

Copy your lab4 ex. 4 code into lab5_ex1.asm

Modify your assn 3 code as needed to transform it into a subroutine, and add it to the file - be sure to adhere to the structure described above, and make sure it issues a newline at the end.

Finally, invoke your new subroutine from inside the display loop.

The end result will be 10 lines of output, representing the first 10 powers of 2 as 16-bit binary numbers (displayed, of course, as ascii 1's and 0's).

Exercise 2

Write the inverse of a binary printing subroutine - that is, write a binary reading subroutine:

First, the subroutine will prompt the user to enter a 16-bit 2's complement binary number: the user will enter 'b' followed by exactly sixteen 1's and 0's: e.g. "b0001001000110100" (*no spaces on input*)

Your subroutine should do the following:

1. The user enters a binary number as a sequence of 17 ascii characters: b0010010001101000
2. The result of (1) is transformed into a single 16-bit value, which is stored in R2.

Your "main" can now invoke the subroutine you built in exercise 1 to print the value of R2 back out to the console to check your work.

"Binary read" Algorithm:

```
R2          <-- 0
counter     <-- 16
R0          <-- get input from user (the initial 'b') and do nothing with it
do
{
    ; that's your job :)
    ; HINT: the 4-bit binary number b1011 is (b101 * #2 + 1)
} while ( counter > 0 );
```

Once again, you might want to consult the [AL control structure tutorials](#)

Exercise 3

Enhance exercise 2 so that it now performs some input validation:

- If the first character entered is not 'b', the program should output an error message and go back to the beginning
- After that (*i.e. inside the loop*), if a SPACE is ever entered, the program should only echo it and continue - i.e. spaces are accepted but ignored in the conversion algorithm.
- If any character other than '1', '0' or SPACE is entered after the initial 'b', the program should output an error message and ask for a valid character - i.e. it should keep everything received so far, and keep looping until it gets a valid '0', '1' or space, and then continue.

Exercise 4: Just read this

From now on, **all** of your programs will consist of a simple test harness invoking one or more subroutines in which you will implement the assigned task – so make sure you have completely mastered the art of dividing your program up into these “self-contained” modules.

Each subroutine should perform a single task, and have CLEAR and EXPLICIT comments describing what it does, what input is required in which registers, and what will be returned in which register.

A subroutine may invoke another subroutine - but only if you are very, very careful!!

We will let you know when we want you to do this.

Make sure you understand and always employ basic “register hygiene” – i.e. backup and restore **ONLY** those registers that are modified by the subroutine for internal purposes - and always backup and restore R7.

3.3 Submission

Demo your lab exercises to your TA ***before you leave lab.***

If you are unable to complete all exercises in lab, show your TA how far you got, and request permission to complete it after lab.

Your TA will usually give you partial credit for what you have done, and allow you to complete & demo the rest later for full credit, so long as you have worked at it seriously.

When you're done, demo it to any of the TAs or instructors in office hours ***before your next lab.***

Office hours are posted on Piazza, under the "Staff" tab.

4 So what do I know now?

... Pretty much everything you need to write real Assembly Language programs :)